

Lagrange Problems

© 2006 by Mike May, S.J.- maymk@slu.edu

with an aside on solving systems of equations

(Section 15.3)

>

In prior semesters, students commented that they were having trouble with the homework from the section on Lagrange multipliers. This worksheet shows how to use Maple to check your work on those problems.

First we look at the slick way that Maple gives an answer, but hides all the details. It uses the solve command to solve a system of equations. Then we step through the work needed to solve a system by hand.

Finding constrained extrema of a function in two variables.

Problem #6) Find the extrema of $f(x,y) = x*y$ subject to the constraint that $g(x,y) = 4*x^2 + y^2 - 8 = 0$.

```
> f := (x, y) -> x*y;  
g := (x, y) -> 4*x^2 + y^2 - 8;  
f := (x, y) -> x*y  
g := (x, y) -> 4*x^2 + y^2 - 8
```

The method we have been studying produces equations relating the partial derivatives of f and g with respect to x and with respect to y .

```
> L[x] := diff(f(x,y), x) - lambda*diff(g(x,y), x);  
L[y] := diff(f(x,y), y) - lambda*diff(g(x,y), y);  
L_x := y - 8*lambda  
L_y := x - 2*lambda
```

We now solve the system of equations made up of the constraint and the relations on the partial derivatives.

```
> solve({g(x,y), L[x], L[y]}, {x, y, lambda});  
{lambda = 1/4, y = 2, x = 1}, {lambda = 1/4, y = -2, x = -1}, {lambda = -1/4, y = 2, x = -1}, {lambda = -1/4, y =  
-2, x = 1}
```

This gives four solutions that we can plug in to find which is a maximum and which is a minimum.

```
> 'f(1,2)' = f(1,2);  
'f(1,-2)' = f(1,-2);  
'f(-1,2)' = f(-1,2);  
'f(-1,-2)' = f(-1,-2);  
f(1,2) = 2  
f(1,-2) = -2  
f(-1,2) = -2  
f(-1,-2) = 2
```

Thus we have maxima at (1,2) and (-1, -2) and minima at (1,-2) and(-1,2).

Finding constrained extrema of a function in three variables.

This technique easily extends to problems in more variables and with more constraints.

Problem # 11) Find the extrema of $f(x, y, z) = x + y + z$, subject to the constraints

$x^2 + y^2 + z^2 = 1$ and $x = y + 1$.

```
> f := (x, y, z) -> x + y + z;
g := (x, y, z) -> x^2 + y^2 + z^2 - 1;
h := (x, y, z) -> x - y - 1;
f := (x, y, z) -> x + y + z
g := (x, y, z) -> x^2 + y^2 + z^2 - 1
h := (x, y, z) -> x - y - 1
```

Note that since there are two constraints in this problem, we have two lambda variables in the system of equations, so that we have a system of five equations in five variables.

```
> L[x] := diff(f(x,y,z),x) + lambda[1]*diff(g(x,y,z),x) +
lambda[2]*diff(h(x,y,z),x);
L[y] := diff(f(x,y,z),y) + lambda[1]*diff(g(x,y,z),y) +
lambda[2]*diff(h(x,y,z),y);
L[z] := diff(f(x,y,z),z) + lambda[1]*diff(g(x,y,z),z) +
lambda[2]*diff(h(x,y,z),z);
Lx := 1 + 2 lambda_1 x + lambda_2
Ly := 1 + 2 lambda_1 y - lambda_2
Lz := 1 + 2 lambda_1 z
```

```
> solve({g(x,y,z), h(x,y,z), L[x], L[y], L[z]},
{x,y,z,lambda[1], lambda[2]});
{y = -1/3 RootOf(-3 + 2_Z^2) - 1/2, z = -1/3 RootOf(-3 + 2_Z^2), x = -1/3 RootOf(
-3 + 2_Z^2) + 1/2, lambda_1 = RootOf(-3 + 2_Z^2), lambda_2 = -RootOf(-3 + 2_Z^2)}
```

Note that $\text{RootOf}(-3 + 2_Z^2) = \pm \sqrt{\frac{3}{2}}$, since the expression $\text{RootOf}(-3 + 2_Z^2)$ literally means the solutions of the equation $-3 + 2z^2 = 0$, or $2z^2 = 3$. Therefore the solutions of the system of equations are $(\frac{1}{2} - \frac{1}{\sqrt{6}}, -\frac{1}{2} - \frac{1}{\sqrt{6}}, -\frac{1}{\sqrt{6}})$ and $(\frac{1}{2} + \frac{1}{\sqrt{6}}, -\frac{1}{2} + \frac{1}{\sqrt{6}}, \frac{1}{\sqrt{6}})$. The first is a minimum, and the second is a maximum.

```
> 'f(1/2+1/sqrt(6), -1/2+1/sqrt(6), 1/sqrt(6))' =
f(1/2+1/sqrt(6), -1/2+1/sqrt(6), 1/sqrt(6));
'f(1/2-1/sqrt(6), -1/2-1/sqrt(6), -1/sqrt(6))' =
f(1/2-1/sqrt(6), -1/2-1/sqrt(6), -1/sqrt(6));
f(1/2 + 1/sqrt(6), -1/2 + 1/sqrt(6), 1/sqrt(6)) = 1/2 sqrt(6)
f(1/2 - 1/sqrt(6), -1/2 - 1/sqrt(6), -1/sqrt(6)) = -1/2 sqrt(6)
```

Solving systems of equations by eliminating variables.

As stated above, this method gives answers but does not show how to solve the system of equations. We now look at the second example again, this time breaking down the problem of solving the system of equations into steps that could be reproduced by hand. We solve the system by repeatedly solving an equation for a variable and using the subs (for substitute) command to eliminate that variable.

Start by collecting the system of equations to be solved.

```
> eq[1] := 0 = g(x,y,z); eq[2] := 0 = h(x,y,z);
   eq[3] := 0 = L[x]; eq[4] := 0 = L[y]; eq[5] := 0 = L[z];
      eq1 := 0 = x2 + y2 + z2 - 1
      eq2 := 0 = x - y - 1
      eq3 := 0 = 1 + 2 λ1 x + λ2
      eq4 := 0 = 1 + 2 λ1 y - λ2
      eq5 := 0 = 1 + 2 λ1 z
```

The first substitution we see comes from equation 2, where we see that $x = y + 1$. We make that substitution in all five equations.

```
> for i from 1 to 5 do
   eq[i] := subs(x = y+1, eq[i]);
od;
      eq1 := 0 = (y + 1)2 + y2 + z2 - 1
      eq2 := 0 = 0
      eq3 := 0 = 1 + 2 λ1 (y + 1) + λ2
      eq4 := 0 = 1 + 2 λ1 y - λ2
      eq5 := 0 = 1 + 2 λ1 z
```

Notice that the substitution wipes out the equation that it came from. The next substitution comes from equation 5, with $\text{lambda}[1] = -1/(2*z)$.

```
> for i from 1 to 5 do
   eq[i] := subs(lambda[1] = -1/(2*z), eq[i]);
od;
      eq1 := 0 = (y + 1)2 + y2 + z2 - 1
      eq2 := 0 = 0
      eq3 := 0 = 1 -  $\frac{y+1}{z}$  + λ2
      eq4 := 0 = 1 -  $\frac{y}{z}$  - λ2
      eq5 := 0 = 0
```

Now we use equation 4 to eliminate lambda[2].

```
> for i from 1 to 5 do
   eq[i] := subs(lambda[2] = 1 - y/z, eq[i]);
od;
```

$$eq_1 := 0 = (y + 1)^2 + y^2 + z^2 - 1$$

$$eq_2 := 0 = 0$$

$$eq_3 := 0 = 2 - \frac{y + 1}{z} - \frac{y}{z}$$

$$eq_4 := 0 = 0$$

$$eq_5 := 0 = 0$$

Solving equation 3 for z gives $z = (2*y + 1)/2$.

```
> solve(eq[3], z);
```

$$y + \frac{1}{2}$$

(3.1)

Substitute $z = (2*y + 1)/2$ into each equation

```
> for i from 1 to 5 do
  eq[i] := simplify(subs(z = (2*y+1)/2, eq[i]));
od;
```

$$eq_1 := 0 = 3y^2 + 3y + \frac{1}{4}$$

$$eq_2 := 0 = 0$$

$$eq_3 := 0 = 0$$

$$eq_4 := 0 = 0$$

$$eq_5 := 0 = 0$$

We are now ready to solve for y.

```
> ansy := [solve(eq[1], y)];
```

$$ansy := \left[-\frac{1}{2} - \frac{1}{6}\sqrt{6}, -\frac{1}{2} + \frac{1}{6}\sqrt{6} \right]$$

```
>
```

```
>
```

Recall that $x = y + 1$ and $z = y + 1/2$. That gives us the same points as we obtained above.

Exercises

Use Maple to solve the following problems. On at least one of the problems solve the system of equations both ways as shown above. Be sure to label your answers. Note that the last two problems require that you also check the interior of a region for extrema since they involve inequality constraints.

1) Find the extrema of $f(x, y) = x^2 + y$, subject to the constraint $x^2 - y^2 = 1$.

```
>
```

2) Find the extrema of $f(x, y, z) = x + 3*y + 5*z$, subject to the constraint $x^2 + y^2 + z^2 = 1$.

```
>
```

3) Find the extrema of $f(x, y) = xy$, subject to the constraint $x^2 + 2y^2 \leq 1$.

```
>
```

4) Find the extrema of $f(x, y) = x^3 - y^2$, subject to the constraint $x^2 + y^2 \leq 1$

```
>
```